
E. Chloroprene PBPK Model Equations 
 
#Chloroprene PBPK Model 
#Translated into R from the acslX model presented in Yang et al. 2012 
#By Jerry Campbell 
#The full model code, including the scripts for running the model, is available from the authors.  
 
States  = { 
AI , 
AX , 
AM , 
AMLU , 
AMK , 
ALU , 
AL , 
AK , 
AS , 
AR , 
AF  
}; 
 
Outputs = { 
MASBAL , 
CLU , 
CL , 
CK , 
CS , 
CR , 
CF , 
CVLUM , 
ppm , 
AMP , 
AMPLU , 
AMPK  , 
cvl , 
qcbal , 
vbal 
}; 
 
Inputs = {EXPPULSE}; 
 
#BODY WEIGHT (kg) 
      BW = 0.03 ;    # Body weight (kg) 
 



 
#SPECIAL FLOW RATES 

QPC = 29.1 ;     # Unscaled Alveolar Vent (L/h/kg^0.75) 
QCC = 20.1 ;     # Unscaled Cardiac Output (L/h/kg^0.75) 

 
#FRACTIONAL BLOOD FLOWS TO TISSUES 

QLC = 0.161 ;   # Flow to Liver as % Cardiac Output (unitless) 
QFC = 0.07  ;   # Flow to Fat as % Cardiac Output (unitless) 
QSC = 0.159 ;   # Flow to Slow as % Cardiac Output (unitless) 
QKC = 0.09  ;   # Flow to Kidney as % Cardiac Output (unitless) 

   
#FRACTIONAL VOLUMES OF TISSUES 

VLC = 0.055 ;      # Volume Liver as % Body Weight (unitless) 
VLUC = 0.0073 ;   # Volume Lung as % Body Weight (unitless) 
VFC = 0.1  ;          # Volume Fat as % Body Weight (unitless) 
VRC = 0.08098 ;      # Volume Rapid Perfused as % Body Weight (unitless) 
VSC = 0.384  ;       # Volume Slow Perfused as % Body Weight (unitless) 
VKC = 0.0167 ;      # Volume Kidney as % Body Weight  (unitless) 

   
#PARTITION COEFFICIENTS PARENT 

PL = 1.26 ;    # Liver/Blood Partition Coefficient (unitless) 
PLU = 2.38 ;    # Lung/Blood Partition Coefficient (unitless) 
PF = 17.35 ;    # Fat/Blood Partition Coefficient (unitless) 
PS = 0.59  ;   # Slow/Blood Partition Coefficient (unitless) 
PR = 1.76 ;    # Rapid/Blood Partition Coefficient (unitless) 
PB = 7.83 ;    # Blood/Air Partition Coefficient (unitless) 
PK = 1.76 ;    # Kidney/Blood Partition Coefficient (unitless) 

 
#KINETIC CONSTANTS 

MW = 88.5  ;   # Molecular weight (g/mol) 
 

# Metabolism in Liver 
VMAXC = 7.95  ;   # Scaled VMax for Oxidative Pathway:Liver (mg/h/BW^0.75) 
KM = 0.041  ;   # Km for Oxidative Pathway:Liver (mg/L) 
 

# Metabolism in Lung 
 VMAXCLU = 0.18;     # Scaled VMax for Oxidative Pathway:Lung (mg/h/BW^0.75) 
KMLU = 0.26;         # Km for Oxidative Pathway:Lung (mg/L) 
KFLUC = 0.0;           # Pseudo-first order clearance in lung (Km unidentifiable) (L/hr/BW^0.75) 

 
# Metabolism in Kidney  

VMAXCKid = 0.0 ;    # Scaled VMax for Oxidative Pathway:Kidney (mg/h/BW^0.75) 
KMKD = 0.0  ;   # Km for Oxidative Pathway :Kidney   
KFKIC = 0.079;     # Pseudo-first order clearance in kidney (Km unidentifiable) (L/hr/BW^0.75) 



 
#DOSING INFORMATION  

TSTOP = 7.0 ;     # Dosing stop time   
CONC = 13.0  ;   # Initial concentration (ppm)  

 
Dynamics { 
# Scaled parameters 

QC = QCC*pow(BW,0.75)  ;       #Cardiac output 
QP = QPC*pow(BW,0.75)  ;       #Alveolar ventilation 
QL = QLC*QC    ;               #Liver blood flow 
QF = QFC*QC    ;               #Fat blood flow 
QS = QSC*QC    ;               #Slowly-perfused tissue blood flow 
QK = QKC*QC    ;               #Kidney tissue blood flow  
QRC = 1-QLC-QKC-QFC-QSC  ;      #Rapidly Perfused tissues 
QR = QRC*QC  ;               #Rapidly-perf tissue blood flow 

 
VL = VLC*BW      ;             #Liver volume 
VLU = VLUC*BW     ;             #Lung volume 
VF = VFC*BW      ;             #Fat tissue volume 
VS = VSC*BW      ;             #Slowly-perfused tissue volume 
VR = VRC*BW      ;             #Richly-perfused tissue volume 
VK = VKC*BW      ;             #kidney tissue volume  

 
ROBC = 1 - VLC - VLUC - VFC - VSC - VRC - VKC ;  #Rest of body un-perfused tissue for MC sims 

 
# METABOLISM 

VMAX = VMAXC*pow(BW,0.75)  ;       #Maximum rate of metabolism-Liver (mg/hr/kg-BW) 
VMAXLU = VMAXCLU*pow(BW,0.75) ;      #Maximum rate of metabolism-Lung (mg/hr/kg-BW) 
KFLU = KFLUC*BW ; 
VMAXKD = VMAXCKid*pow(BW,0.75)  ;    #Maximum rate of metabolism-Kidney (mg/hr/kg-BW) 
KFKI = KFKIC*BW ; 

 
# Exposure Control (mg/L) 

CIX = CONC*MW/24450 ;                 
CI = CIX *EXPPULSE  ; 

 
# Tissue Venous Concentrations (mg/L) 

CVLU = ALU/(VLU*PLU) ; 
CVL = AL/(VL*PL) ; 
CVK = AK/(VK*PK) ; 
CVS = AS/(VS*PS) ; 
CVR = AR/(VR*PR) ; 
CVF = AF/(VF*PF) ; 

 



# Concentration in Pulmonary/Arterial and venous blood Compartments (mg/L) 
CPU = (QP*CI+(QF*CVF + QL*CVL + QS*CVS + QR*CVR + QK*CVK))/(QP/PB+QC)  ; 
CX = CPU/PB ; 
CV = (QF*CVF + QL*CVL + QS*CVS + QR*CVR + QK*CVK)/QC ; 
CPUM = CPU*1000/MW ; 

 
RAI = QP*CI ; 
dt(AI) = RAI ; 
RAX = QP*CX ; 
dt(AX) = RAX ; 

 
# Amount metabolized in Liver (mg) 

RAM = VMAX*CVL/(KM + CVL) ; 
dt(AM) = RAM ; 

 
# Amount metabolized in Lung (mg) 

RAMLU = VMAXLU*CVLU/(KMLU + CVLU) + KFLU*CVLU ; 
dt(AMLU) = RAMLU ; 

  
# Amount metabolized in Kidney (mg) 

RAMK = VMAXKD*CVK/(KMKD + CVK) + KFKI*CVK ; 
dt(AMK) = RAMK ; 

 
# Amount in Lung Compartment (mg)      

RALU = QC*(CPU-CVLU) - RAMLU ; 
dt(ALU) = RALU ; 

 
# Amount in Liver Compartment (mg) 

RAL = QL*(CVLU-CVL) - RAM ; 
dt(AL) = RAL ; 

 
# Amount in Kidney Compartment (mg) 

RAK = QK*(CVLU-CVK) - RAMK ; 
dt(AK) = RAK ; 

 
# Amount in Slowly Perfused Tissues (mg) 

RAS = QS*(CVLU - CVS) ; 
dt(AS) = RAS ; 

 
# Amount in Rapidly Perfused Tissues (mg) 

RAR = QR*(CVLU -CVR) ; 
dt(AR) = RAR ; 

 
# Amount in Fat Compartment (mg) 



      RAF = QF*(CVLU - CVF) ; 
      dt(AF) = RAF ; 
 
} # End of Dynamics 
 
 
CalcOutputs { 
# Mass-balance 
 MASBAL = AI - AX - (AL+AM+AMLU+ALU+AK+AMK+AS+AR+AF) ; 
 
#Tissue Concentrations (mg/L)    

CLU = ALU/VLU ; 
CL = AL/VL ; 
CK = AK/VK ; 
CS = AS/VS ; 
CR = AR/VR ; 
CF = AF/VF ; 

#Concentrations for plots 
 CVLUM = CVLU*1000/MW ;  #(umol/L) 
#Dose metrics 

ppm = CONC ; 
AMP = ((AM*1000/MW)/(VL*1000))/(TSTOP/24) ; 
AMPLU = ((AMLU*1000/MW)/(VLU*1000))/(TSTOP/24) ; 
AMPK = ((AMK*1000/MW)/(VK*1000))/(TSTOP/24) ; 

    
cvl = CVL ; 
    
#Blood Flow balance 

qcbal = QC - QL - QF - QS - QK - QR ;   
#Tissue Volume balance 

vbal = BW*(1-ROBC) - VL - VLU - VF - VS - VK - VR ; 
    
} # End of CalcOutputs 
 
End 


